

# 15–150: Principles of Functional Programming

## *Some Notes on Structural Induction*

Michael Erdmann\*

Spring 2025

[This is a version of the notes that first appeared in Lecture 4, there specialized to integers. Here now they are polymorphic.]

These notes provide a brief introduction to structural induction for proving properties of SML programs. We assume that the reader is already familiar with SML and the notes on evaluation and natural number induction for pure SML programs.

Structural inductions in SML often arise as inductions over the structure of values defined by `datatype` declarations. Most `datatype` declarations give rise to an induction principle which may be used to prove properties of recursive functions with arguments of the given type.

**Notation Reminder:** We write  $e \xrightarrow{k} e'$  (or  $e \Rightarrow^k e'$ ) for a computation of  $k$  steps,  $e \Rightarrow e'$  (or  $e \Rightarrow^* e'$ ) for a computation of any number of steps (including 0),  $e \hookrightarrow v$  for a complete computation of  $e$  to a value  $v$ , and  $n = m$  or  $e = e'$  for mathematical equality.

We say that two expressions  $e$  and  $e'$  are *extensionally equivalent*, and write  $e \cong e'$ , whenever one of the following is true: (i) evaluation of  $e$  produces the same value as does evaluation of  $e'$ , or (ii) evaluation of  $e$  raises the same exception as does evaluation of  $e'$ , or (iii) evaluation of  $e$  and evaluation of  $e'$  both loop forever. In other words, evaluation of  $e$  appears to behave just as does evaluation of  $e'$ . We say that an expression *has no value* if its evaluation either raises an exception or loops forever.

Observe: If  $e_1 \Rightarrow e$  and  $e_2 \Rightarrow e$ , then  $e_1 \cong e_2$ . In other words, “reduction implies equivalence”. CAUTION: The opposite implication need not hold: two expressions can be equivalent without one reducing to the other. For instance,  $(1 + 9) \cong (2 * 5)$ , but neither expression reduces to the other.

In proofs that establish equivalences, we frequently justify an equivalence  $e_1 \cong e_2$  by writing something like `[step, 2nd clause of f]`. This means that either  $e_1 \Rightarrow e_2$  or  $e_2 \Rightarrow e_1$ , as established by the function clause being cited. (Observe that such a reduction does indeed imply  $e_1 \cong e_2$ , by the previous paragraph.)

---

\*Modified from a draft by Frank Pfenning.

## 1 Proof By Cases

A very simple form of “structural induction” arises if the datatype declaration is not recursive, but provides a finite number of data constructors. For such datatypes we can prove theorems by cases, which may also be viewed as an induction with only base cases. As an example, consider the declaration

```
datatype PrimColor = Red | Green | Blue
```

We can now prove properties of all primitive colors by distinguishing the cases of `Red`, `Green`, and `Blue`.

Another form for proof by cases arises for the Booleans, since there is a pervasive definition

```
datatype bool = true | false
```

For example, it is easy to see that

$$(\text{if } e \text{ then } e' \text{ else } e') \not\equiv e'$$

since, for instance, evaluation of  $e$  might loop forever, whereas  $e'$  could reduce to a value. However, if  $e$  reduces to a value, then the two expressions are extensionally equivalent, as we now prove:

**Theorem 1** *For every expression  $e$  (of type `bool`) such that  $e \hookrightarrow v$  for some  $v$ , and for every  $e'$ ,*

$$(\text{if } e \text{ then } e' \text{ else } e') \cong e'$$

**Proof:** By cases on the value of  $e$ .

$$\begin{aligned} & \text{if } e \text{ then } e' \text{ else } e' \\ \implies & \text{if } v \text{ then } e' \text{ else } e' \quad [\text{by assumption on } e] \end{aligned}$$

Now either  $v = \text{true}$  or  $v = \text{false}$  by cases on the structure of `bool`. In either case, the expression above reduces to  $e'$ .  $\square$

## 2 Structural Induction on Lists

The pervasive type of `'a list` is defined in SML by:

```
datatype 'a list = nil | :: of 'a * 'a list
infixr ::
```

The declaration `infixr ::` changes the lexical status of the constructor `::` to be a right-associative infix operator. That is, `1::2::3::nil` should be read as `1::(2::(3::nil))` which in turn would correspond to `::(1, ::(2, ::(3, nil)))` if `::` had not been declared infix. SML provides an alternative syntax for lists, as follows:

$$\begin{aligned} [] &= \text{nil} \\ [e_1, e_2, \dots, e_n] &= e_1 :: (e_2 :: (\dots (e_n :: \text{nil}))) \end{aligned}$$

The recursive nature of the declaration of `list` means that the corresponding induction principle is not just a proof by cases. It reads:

<b>If:</b>	1. a property holds for the empty list <code>nil</code> <b>and</b> 2. whenever the property holds for a value $l$ of type <code>t list</code> , it also holds for $v :: l$ (for any value $v$ of type $t$ ),
<b>then:</b>	the property holds for all values of type <code>t list</code> .

As a very simple example, consider the definition of a function to append two lists:

```
(* @ : 'a list * 'a list -> 'a list *)
fun @ (nil, k) = k
  | @ (x::l, k) = x :: @(l,k)
infixr @
```

Appending two lists that are values always reduces to a value in SML. While this may seem trivial, it is actually not the case for some other functional languages such as Haskell in which values may be defined recursively.

**Lemma 2** *The function  $\text{@}$  is total. In other words, for any values  $l$  and  $k$  of type `t list` (with  $t$  an arbitrary type),  $l @ k \hookrightarrow v$  for some  $v$ .*

**Proof:** By structural induction on  $l$ .

**Base Case:**  $l = \text{nil}$ .

We need to show that for any value  $k : t \text{ list}$ ,  $\text{nil} @ k$  reduces to a value. Well,

$$\text{nil} @ k \implies k \quad [\text{by the first clause of } @]$$

**Inductive Case:**  $l = x :: l'$  for some values  $x : t$  and  $l' : t \text{ list}$ .

Induction hypothesis: Assume, for any value  $k : t \text{ list}$ ,  $l' @ k \hookrightarrow v'$  for some  $v'$ .

We need to show that, for any value  $k : t \text{ list}$ ,  $l @ k \hookrightarrow v$  for some  $v$ .

Evaluating code, we see that:

$$\begin{aligned} & (x :: l') @ k \\ \implies & x :: (l' @ k) \quad [\text{by the second clause of } @] \\ \implies & x :: v' \quad [\text{by induction hypothesis on } l'] \\ = & v \quad (\text{writing } v \text{ for the value } x :: v') \end{aligned}$$

By the structural induction principle for lists, this completes the proof.  $\square$

One can also prove that  $l @ k$  takes  $O(|l|)$  steps, where  $|l|$  is the length of the list  $l$ . From this observation one can see that  $(l @ k) @ m$  takes  $O(2|l| + |k|)$  steps, while  $l @ (k @ m)$  takes only  $O(|l| + |k|)$  steps. This is the basis for a number of simple efficiency improvements one can make in SML programs. It is formalized in the following lemma.

**Lemma 3** *For any values  $l_1$ ,  $l_2$ , and  $l_3$  of type `t list` (with  $t$  an arbitrary type),*

$$(l_1 @ l_2) @ l_3 \cong l_1 @ (l_2 @ l_3)$$

**Proof:** We reformulate this slightly to simplify the presentation of the proof:

$$(l_1 @ l_2) @ l_3 \cong l_{12} @ l_3 \cong l_{123} \quad \text{iff} \quad l_1 @ (l_2 @ l_3) \cong l_1 @ l_{23} \cong l_{123}$$

The proof is by structural induction on  $l_1$ .

**Base Case:**  $l_1 = \text{nil}$ .

We need to show that for any pair of list values  $l_2$  and  $l_3$  of type  $\mathbf{t} \text{ list}$ ,  $(\text{nil} @ l_2) @ l_3$  is extensionally equivalent to a value  $l_{23}$  iff  $\text{nil} @ (l_2 @ l_3)$  is extensionally equivalent to the value  $l_{23}$ . Well,

$$\begin{aligned} & (\text{nil} @ l_2) @ l_3 \\ \cong & l_2 @ l_3 && [\text{step, first clause of } @] \\ \cong & l_{23} && [\text{for some value } l_{23}, \text{ by totality of } @, \text{i.e., Lemma 2}] \end{aligned}$$

and

$$\begin{aligned} & \text{nil} @ (l_2 @ l_3) \\ \cong & \text{nil} @ l_{23} && [\text{by totality of } @] \\ \cong & l_{23} && [\text{step, first clause of } @] \end{aligned}$$

**Inductive Case:**  $l_1 = x :: l'_1$  for some values  $x : \mathbf{t}$  and  $l'_1 : \mathbf{t} \text{ list}$ .

Induction hypothesis: Assume, for any values  $l_2$  and  $l_3$  of type  $\mathbf{t} \text{ list}$ ,

$$(l'_1 @ l_2) @ l_3 \cong l'_{12} @ l_3 \cong l'_{123} \quad \text{iff} \quad l'_1 @ (l_2 @ l_3) \cong l'_1 @ l_{23} \cong l'_{123}$$

We need to show that, for any values  $l_2$  and  $l_3$  of type  $\mathbf{t} \text{ list}$ ,

$$(l_1 @ l_2) @ l_3 \cong l_{12} @ l_3 \cong l_{123} \quad \text{iff} \quad l_1 @ (l_2 @ l_3) \cong l_1 @ l_{23} \cong l_{123}$$

Evaluating code, for the left expression we obtain:

$$\begin{aligned} & ((x :: l'_1) @ l_2) @ l_3 \\ \cong & (x :: (l'_1 @ l_2)) @ l_3 && [\text{step, second clause of } @] \\ \cong & (x :: l'_{12}) @ l_3 && [\text{for some value } l'_{12}, \text{ by totality of } @] \\ \cong & x :: (l'_{12} @ l_3) && [\text{step, second clause of } @] \\ \cong & x :: l'_{123} && [\text{for some value } l'_{123}, \text{ by totality of } @] \end{aligned}$$

For the right expression we obtain:

$$\begin{aligned} & (x :: l'_1) @ (l_2 @ l_3) \\ \cong & (x :: l'_1) @ l_{23} && [\text{for some value } l_{23}, \text{ by totality of } @] \\ \cong & x :: (l'_1 @ l_{23}) && [\text{step, second clause of } @] \\ \cong & x :: l'_{123} && [\text{by the induction hypothesis on } l'_1] \end{aligned}$$

By the structural induction principle for lists, this completes the proof.  $\square$

We actually have the stronger and often useful result that  $@$  is associative even for expressions which are not necessarily values, assuming sequential code evaluation. This holds even under extensions by arbitrary effects, since in  $e_1 @ (e_2 @ e_3)$  and  $(e_1 @ e_2) @ e_3$ , the expressions  $e_1$ ,  $e_2$  and  $e_3$  are evaluated in the same order, with all  $@$  computations in between reducing to values whenever the arguments are values.

**Lemma 4** For arbitrary expressions  $e_1$ ,  $e_2$  and  $e_3$  (of the same list type),

$$(e_1 @ e_2) @ e_3 \cong e_1 @ (e_2 @ e_3)$$

**Proof:** By straightforward computation and Lemma 3:

$$\begin{aligned} & (e_1 @ e_2) @ e_3 \\ \implies & (l_1 @ e_2) @ e_3 \quad \text{or } e_1 \text{ has no value} \\ \implies & (l_1 @ l_2) @ e_3 \quad \text{or } e_2 \text{ has no value} \\ \implies & l_{12} @ e_3 \quad [\text{by totality of } @] \\ \implies & l_{12} @ l_3 \quad \text{or } e_3 \text{ has no value} \\ \implies & l_{123} \quad [\text{by totality of } @] \end{aligned}$$

For the right-hand side we compute:

$$\begin{aligned} & e_1 @ (e_2 @ e_3) \\ \implies & l_1 @ (e_2 @ e_3) \quad \text{or } e_1 \text{ has no value} \\ \implies & l_1 @ (l_2 @ e_3) \quad \text{or } e_2 \text{ has no value} \\ \implies & l_1 @ (l_2 @ l_3) \quad \text{or } e_3 \text{ has no value} \\ \implies & l_1 @ l_{23} \quad [\text{by totality of } @] \\ \implies & l'_{123} \quad [\text{by totality of } @] \\ \cong & l_{123} \quad [\text{by proof of Lemma 3}] \end{aligned}$$

□

### 3 Structural Induction on Other Types

As an example for structural induction over other types we use binary trees in which the leaves carry integer data:

```
datatype 'a tree = Leaf of 'a | Node of 'a tree * 'a tree
```

The structural induction principle for these types of trees then reads:

**If:**

1. a property holds for every leaf `Leaf(x)`, with  $x$  a value of type  $s$ ,
- and
2. whenever the property holds for values  $t_1$  and  $t_2$  of type `s tree`,  
it also holds for `Node(t1, t2)`,

**then:** the property holds for all values of type `s tree`.

The following function is inefficient, since the elements of `flatten t1` may end up being copied many times when the result lists are appended.

```
(* flatten : 'a tree -> 'a list
  REQUIRES: true
  ENSURES: flatten(t) computes the inorder traversal of the leaf values.
*)
fun flatten (Leaf(x)) = [x]
  | flatten (Node(t1,t2)) = flatten t1 @ flatten t2
```

A more efficient alternative introduces an accumulator argument.

```
(* flatten2 : 'a tree * 'a list -> 'a list
  REQUIRES: true
  ENSURES: flatten2 (t, acc)  $\cong$  flatten (t) @ acc
*)
fun flatten2 (Leaf(x), acc) = x::acc
  | flatten2 (Node(t1,t2), acc) =
    flatten2 (t1, flatten2 (t2, acc))

(* flatten' : 'a tree -> 'a list *)
fun flatten' (t) = flatten2 (t, nil)
```

We would like to prove that `flatten` and `flatten'` define the same function. In order to do that, we need to prove a lemma about `flatten2`, which requires a generalization of the induction hypothesis: We cannot prove directly by induction that `flatten2(t, nil)  $\cong$  flatten(t)` since recursive calls in `flatten2` have a more general structure. The case of a leaf provides a clue about the proper generalization.

**Lemma 5** *For any values  $t$  of type `s tree` and  $acc$  of type `s list` (with  $s$  an arbitrary type),*

```
flatten2(t, acc)  $\cong$  flatten(t) @ acc
```

**Proof:** By structural induction on  $t$ .

**Base Case:**  $t = \text{Leaf}(x)$ , for some value  $x : \mathbf{s}$ .

We need to show that, for all values  $acc$  of type  $\mathbf{s} \text{ list}$ ,  
 $\text{flatten2}(\text{Leaf}(x), acc) \cong \text{flatten}(\text{Leaf}(x)) @ acc$ .

Showing:

$$\begin{aligned}
& \text{flatten2}(\text{Leaf}(x), acc) \\
& \cong x :: acc && [\text{step, first clause of } \text{flatten2}] \\
& \cong x :: (\text{nil} @ acc) && [\text{step, first clause of } @] \\
& \cong (x :: \text{nil}) @ acc && [\text{step, second clause of } @] \\
& = [x] @ acc \\
& \cong \text{flatten}(\text{Leaf}(x)) @ acc && [\text{step, first clause of } \text{flatten}]
\end{aligned}$$

**Inductive Case:**  $t = \text{Node}(t_1, t_2)$  for some values  $t_1$  and  $t_2$ , both of type  $\mathbf{s} \text{ tree}$ .

Induction hypothesis: Assume that

for any value  $acc$  of type  $\mathbf{s} \text{ list}$ ,  $\text{flatten2}(t_1, acc) \cong \text{flatten}(t_1) @ acc$  and  
for any value  $acc$  of type  $\mathbf{s} \text{ list}$ ,  $\text{flatten2}(t_2, acc) \cong \text{flatten}(t_2) @ acc$ .

We need to show that for any value  $acc$  of type  $\mathbf{s} \text{ list}$ ,  
 $\text{flatten2}(t, acc) \cong \text{flatten}(t) @ acc$ .

Showing:

$$\begin{aligned}
& \text{flatten2}(\text{Node}(t_1, t_2), acc) \\
& \cong \text{flatten2}(t_1, \text{flatten2}(t_2, acc)) && [\text{step, second clause of } \text{flatten2}] \\
& \cong \text{flatten2}(t_1, \text{flatten}(t_2) @ acc) && [\text{by the IH for } t_2] \\
& \cong \text{flatten}(t_1) @ (\text{flatten}(t_2) @ acc) && [\text{by the IH for } t_1 \text{ and totality of both } \\
& & & \text{flatten and } @ \text{ (Lemma 2)}] \\
& \cong (\text{flatten}(t_1) @ \text{flatten}(t_2)) @ acc && [\text{by associativity of } @ \text{ (Lemma 4)}] \\
& \cong \text{flatten}(\text{Node}(t_1, t_2)) @ acc && [\text{step, second clause of } \text{flatten}]
\end{aligned}$$

By the structural induction principle for trees, this completes the proof.  $\square$

Comment: We used totality of `flatten` in the proof above. Imagine how you might prove that.

The following theorem now follows directly:

**Theorem 6** For any value  $t$  of type  $\mathbf{s} \text{ tree}$ ,

$$\text{flatten}'(t) \cong \text{flatten}(t).$$

**Proof:** We compute directly:

$$\begin{aligned}
& \text{flatten}'(t) \\
& \cong \text{flatten2}(t, \text{nil}) && [\text{by definition of } \text{flatten}'] \\
& \cong \text{flatten}(t) @ \text{nil} && [\text{by Lemma 5}] \\
& \cong \text{flatten}(t)
\end{aligned}$$

Where the last equivalence holds by a property of `@` which is left as an exercise.  $\square$

There are also variants of structural induction analogous to strong induction, where we need to apply the induction hypothesis to some subexpression of the given value. We will not go into further details here.